TUNING YOUR ATARI



by LINDA M. SCHREIBER

There's something about music that fascinates kids. Give them a small piano, drums, harmonica, and they will sit for hours creating their own melodies. A few years ago there was a toy piano on the market that contained a tape recorder. This was a big hit with my daughter. Now, she could not only make her own music, but listen to it afterward.

TUNING YOUR ATARI uses this idea. It is a musical game for children. Type it in and run it, and you will see a simple menu. Choice #3 demonstrates the program. Choice #1 allows you to compose a tune, and Choice #2 will play it back. The tones appear to be made by little figures jumping on a bellows.

Above each figure is the letter name of the tone which that bellows will produce. To operate the bellows, press 1, then press any number from 1–8 on the keyboard. Key one corresponds to the low C; eight to high C. When a number is pressed, the character will jump down on the bellows, flapping his arms as the bellows is compressed. Once the tone is played, he bounces back up to his original position. The program can hold up to 100 notes. If your melody is less than 100 notes, press the escape key and the menu will reappear on the screen. Press #2 to hear your melody.

Young children will enjoy this program just to see the characters jump up and down while they are playing the tunes. Slightly older children will enjoy listening to the tunes that they have created. The letters above the characters do not attract attention, but are a subtle reminder of the names of the notes. After a while, children will begin to associate the letters with the tones of the character. Don't be surprised if you hear your child singing 'A-G-F-G-A-A-A'!

Once again, in this program, we will move the character set out of ROM and into RAM so that we can change some of the characters. In line 70, P1$ should equal h, reverse quotation marks, control D, reverse space, control comma, reverse 1, reverse M, reverse control Q. The characters from K to r are all in reverse. The last character in the string is control period. This string is the machine language subroutine that moves the characters.

Variables Used

P1$	—machine language subroutine
M$	—string holds the melody played
A	—location of the new character set. This value is POKEd into 756 to change to the new character set.
TONE	—line number that starts the tone for the key pressed.
WAIT	—line number for the timing routine.
Q	—no function
CHBS	—first decimal location of the new character set.
X	—no function — used in FOR . . . NEXT loops.
C	—used in READ for new character set, used for value of key pressed, and for position of character.
K	—counter for the note being entered or played.
T	—value of the tone to be played.
TL	—value used in timing loop.
ROUTINE	—the line number that the program goes to when entering the melody, or playing one back.

```
 10 REM TUNING YOUR ATARI
 20 REM BY L.M.SCHREIBER
 30 REM FOR ANTIC OCTOBER 1
    982
 40 DIM M$(101),P1$(20)
 50 GRAPHICS 18:POKE 711,PE
    EK(710):POKE 710,100
 60 A=PEEK(106)-8:POKE 204,
    A:POKE 206,224:REM STOR
    E THE BEGINNING OF NEW
    & OLD CHARACTER SETS
 70 P1$="h" 1MKHPgfNfLJP
    r`":TONE=430:WAIT=500:R
    EM P1$ IS A MACHINE LAN
    GUAGE SUBROUTINE TO MOV
    E THE CHARACTER SET
 80 Q=USR(ADR(P1$)):CHBS=A*
    256:POKE 756,A:REM CHAN
    GE TO THE NEW CHARACTER
     SET
 90 FOR X=CHBS+8 TO CHBS+71
    :READ C:POKE X,C:NEXT X
    :REM CHANGE THE CHARACT
    ERS FROM ! TO $
100 DATA 0,254,124,254,124,
    254,124,254,108,0,254,2
    54,124,254,124,254,40,1
    08,0,254,254,254,124,25
    4
110 DATA 186,40,108,0,254,2
    54,254,254,56,108,56,16
    ,254,56,40,108,0,56,108
    ,56,146,124,56,40
120 DATA 0,0,56,108,56,16,2
    54,56,0,0,0,56,108,56,1
    6,124
130 OPEN #2,4,0,"K:":REM OP
    EN THE KEYBOARD FOR REA
    D
140 POSITION 2,9:? #6;"! !
    ! ! ! ! ! !":POSITION 2
    ,8:? #6;"% % % % % % %
    %":REM THE ! AND % ARE
    THE NEW CHARACTERS
150 POSITION 2,6:? #6;"c d
    e f g a b c":REM PLACE
    THE TONE NAMES
160 K=0:POKE 710,100:REM RE
    STORE THE MENU
170 POSITION 2,0:? #6;"1. P
    LAY KEYBOARD"
180 POSITION 2,2:? #6;"2. R
    EPEAT MELODY"
190 POSITION 2,4:? #6;"3. P
    LAY EXAMPLE"
200 GET #2,C:POKE 710,0:REM
     GET THE KEY PRESSED-RE
    MOVE THE MENU
210 IF C>127 THEN C=C-128:P
    OKE 694,0:REM INVERSE F
    LAG IS ON RESET IT TO N
    ORMAL
220 IF C<49 OR C>52 THEN 16
    0:REM NOT A NUMBER FROM
     1 TO 4
230 C=C-48:REM GET THEN NUM
    BER
240 ON C GOTO 250,540,520,5
    60
250 M$="":REM REMOVE CONTEN
    TS OF THE STRING
260 ROUTINE=260:K=K+1:IF K=
    101 THEN 160:REM ONLY A
    CCEPT 100 NOTES
280 GET #2,C:REM GET THE KE
    Y PRESSED-RETURN TO MEN
    U ON ESCAPE KEY
290 IF C>127 THEN C=C-128:P
    OKE 694,0:REM INVERSE F
    LAG IS ON RESET IT TO N
    ORMAL
300 IF C<49 OR C>56 THEN 16
    0:REM NOT A NUMBER FROM
     1 TO 8
310 C=C-48:M$(K,K)=STR$(C):
    REM GET THEN NUMBER-PUT
     IT IN THE STRING
320 C=C*2:REM OFFSET IT FOR
     THE PROPER POSITION
330 ON C/2 GOSUB 350,360,37
    0,380,390,400,410,420
340 GOTO ROUTINE
350 T=121:GOTO TONE:REM 'C'
360 T=108:GOTO TONE:REM 'D'
370 T=96:GOTO TONE:REM 'E'
380 T=91:GOTO TONE:REM 'F'
390 T=81:GOTO TONE:REM 'G'
400 T=72:GOTO TONE:REM 'A'
410 T=64:GOTO TONE:REM 'B'
420 T=60:REM 'C'
425 REM LINES 430-450 MAKE
    THE CHARACTER APPEAR TO
     PUSH DOWN ON THE BELLO
    W AND MAKE THE TONE
430 TL=10:POSITION C,8:? #6
    ;CHR$(134):POSITION C,9
    :? #6;CHR$(130):SOUND 0
    ,T,10,6:GOSUB WAIT
440 POSITION C,8:? #6;CHR$(
    135):POSITION C,9:? #6;
    CHR$(131):SOUND 0,T,10,
    8:GOSUB WAIT
450 POSITION C,8:? #6;CHR$(
    136):POSITION C,9:? #6;
    CHR$(132):GOSUB WAIT
460 SOUND 0,T,10,10
470 POSITION C,8:? #6;CHR$(
    135):POSITION C,9:? #6;
    CHR$(131):SOUND 0,T,10,
    8:GOSUB WAIT
475 REM LINES 470-490 RETUR
    N THE CHARACTER AND BEL
    LOW TO THE CORRECT POSI
    TION
480 POSITION C,8:? #6;CHR$(
    134):POSITION C,9:? #6;
    CHR$(162):SOUND 0,T,10,
    6:GOSUB WAIT
490 POSITION C,8:? #6;"%":P
    OSITION C,9:? #6;"!":SO
    UND 0,0,0,0:RETURN
500 FOR X=1 TO TL:NEXT X:RE
    TURN :REM TIMING LOOP
510 REM PLAY A SAMPLE TUNE
520 M$="115556544332 21"
530 REM ROUTINE TO PLAY BAC
    K THE MELODY ENTERED
540 ROUTINE=540:K=K+1:IF K<
    =LEN(M$) THEN C=VAL(M$(
    K,K)):GOTO 320:REM KEEP
     PLAYING UNTIL THE END
    OF THE STRING
550 M$(K,K)="0":C=VAL(M$(K,
    K)):M$=M$(1,K-1):GOTO 1
    60
560 CLOSE #2:END
```



Variable checksum = 225145

Line num range	Code	Length
10 - 90	QA	529
100 - 150	SI	504
160 - 250	LD	539
260 - 350	YQ	521
360 - 450	SJ	554
460 - 530	TI	537
540 - 560	QP	187